
CONFIG += c++11

# emit warnings if you use deprecated Qt feature
DEFINES += QT_DEPRECATED_WARNINGS

baseTarget=$${TARGET}

# include base config
include(my.pri)

# define config string
CONFIG(debug, debug|release){
    KT_Configuration=debug
    CharDebug=d
}
else{
    KT_Configuration=release
    CharDebug=
}

# change TARGET
TARGET =$$baseTarget$$CharDebug

# set Root folder by system
unix {
    SYSTEM_STR=linux
    KT_BASE_FOLDER=$$KT_ROOT/$$KT_Configuration/$$SYSTEM_STR
}
macx {
    SYSTEM_STR=mac
    KT_BASE_FOLDER=$$KT_ROOT/$$KT_Configuration/$$SYSTEM_STR
}
win32 {
    SYSTEM_STR=win32
    KT_BASE_FOLDER=$$KT_ROOT/$$KT_Configuration/$$SYSTEM_STR
}

# DESTDIR: define the out exe or dll folder
DESTDIR = $$KT_BASE_FOLDER
OUT_PWD = $$KT_ROOT/build/$${baseTarget}-$${KT_Configuration}-$${SYSTEM_STR}
MOC_DIR = "$$OUT_PWD/.moc"
OBJECTS_DIR = "$$OUT_PWD/.obj"
UI_DIR  = "$$OUT_PWD/.ui"
RCC_DIR = "$$OUT_PWD/.qrc"

# Additional import path used to resolve QML modules in Qt Creator's code model
QML_IMPORT_PATH =

# Additional import path used to resolve QML modules just for Qt Quick Designer
QML_DESIGNER_IMPORT_PATH =

# Default rules for deployment.
qnx: target.path = /tmp/$${TARGET}/bin
else: unix:!android: target.path = /opt/$${TARGET}/bin
!isEmpty(target.path): INSTALLS += target
